1. Три кита языка программирования

	   Любой язык программирования опирается на трех китов: следование, ветвление, повторение (цикл). Я бы добавил сюда еще одного кита - подпрограмму. Подпрограммы у нас впереди, а про остальное поговорим сейчас.

   Всем нам “по жизни” хорошо известно понятие алгоритма. Представьте себе, что в незнакомом городе Вам нужно попасть на вокзал. Если Вы спросите, как туда проехать, то получите ответ типа: “Пройдете до следующего перекрестка и сядете там на автобус №17. Проедете на нем две остановки в сторону центра и пересядете на троллейбус №6 до конечной остановки”. Это типичное следование, т.е. совокупность действий, выполняемых последовательно, друг за другом. Нельзя выполнить очередную инструкцию прежде, чем будет выполнена предыдущая.

   Если возможны варианты в выборе маршрута, то данные Вам инструкции будут содержать ветвление: “Если подойдет автобус №12, то доедете им до вокзала без пересадок, иначе проедете любым транспортом три остановки до универмага и сядете там на автобус №23 ...”. В этом случае Ваши действия зависят от того, какой автобус первым подойдет к остановке.

   Предположим, что Вы хотите дождаться автобуса №12, чтобы с Вашими двумя чемоданами ехать без пересадок. Тогда фактически будет выполнен следующий алгоритм: “Ждать, пока придет автобус №12”. Этот алгоритм содержит повторение. Ждем автобуса. Подходит автобус. Смотрим: №12? Увы, нет. Ждем следующего автобуса. Подходит автобус. Смотрим: №12? Увы, нет. Ждем следующего автобуса... Одни и те же инструкции выполняются несколько раз. Теоретически возможно даже бесконечное число повторений (программисты говорят - зацикливание).

   Ветвление в большинстве языков программирования, в том числе и в Бейсике, реализуется с помощью условного оператора, в котором используются ключевые слова IF, THEN, ELSE, в переводе на русский они означают ЕСЛИ, ТО, ИНАЧЕ.

   В программе на Бейсике оператор ветвления может записываться в одну строку

IF условие THEN ветвь_”то”
- неполное ветвление
или
 

IF условие THEN ветвь_”то” ELSE ветвь_”иначе”
- полное ветвление
и в несколько строк

IF условие THEN
 

   ветвь_”то”
- неполное ветвление
END IF
 

или
 

IF условие THEN
 

   ветвь_”то”
 

ELSE
- полное ветвление
   ветвь_”иначе”
 

END IF
 

   Ветвь “то” исполняется, если при проверке окажется, что условие верно. Если же окажется, что условие неверно, то будет исполнена ветвь “иначе”.

Позже мы познакомимся и с так называемым множественным ветвлением (выбором).

   Операторов цикла в Бейсике несколько. Есть два оператора цикла “повторяй, пока выполняется условие”:

DO
 

   тело_цикла
- проверка условия в конце цикла
LOOP WHILE условие
 

 

 

DO WHILE условие
 

   тело_цикла
- проверка условия в начале цикла
LOOP
 

   Из цикла DO можно выйти досрочно по команде EXIT DO, даже если условие выполнения продолжает оставаться истинным.

   Цикл “для” (цикл с параметром):

FOR параметр = начальное_значение TO конечное_значение [STEP шаг]

   тело_цикла
NEXT параметр

   Заметим, что часть заголовка цикла - STEP шаг - может отсутствовать (поэтому она взята в квадратные скобки!). В этом случае считается, что параметр цикла увеличивается каждый раз на 1.

   Параметр - это имя некоторой переменной величины. Для каждой переменной Quick Basic отводит место в оперативной памяти, где и хранится ее текущее значение.

   Оператор цикла “для” работает следующим образом: при входе в цикл параметру присваивается начальное значение. Проверяется условие “значение параметра не больше конечного значения” при положительном шаге изменения параметра или условие “значение параметра не меньше конечного значения” при отрицательном шаге . Если проверяемое условие истинно, то выполняется тело цикла, затем к значению параметра прибавляется шаг изменения и происходитвозврат к проверке условия. При невыполнении условия управление передается оператору, следующему за признаком конца цикла FOR-NEXT.

   Из этого цикла также можно выйти, не дожидаясь конца повторений с помощью оператора EXIT FOR.

2. Среда программирования Quick Basic 4.5

   Quick Basic – DOS-приложение, т.е. он работает в MS-DOS’е. Запустить Quick Basic можно несколькими способами. Самый простой - войти в каталог Quick Basic’а, выделить курсором файл qb.exe и нажать Enter. Однако это и самый плохой способ. В этом случае придется настраивать Quick Basic на работу с каталогом, в котором хранятся Ваши программы, так как хранить свои программы вмести с файлами Quick Basic’а категорически не рекомендуется!

Второй способ - сделать активным нужный каталог и набрать в командной строке qb.exe (или просто qb) и нажать Enter. Чтобы этот вариант сработал, нужно внести описание пути к файлу qb.exe в autoexec.bat.

   Третий способ - настроить NC так, чтобы Quick Basic запускался автоматически, если выбрать файл с расширением .bas и нажать клавишу Enter. Этот способ требует большего знания NC, чем о ней рассказано в этой книге, но, по мнению автора, он наиболее удобный.

   После запуска программы qb.exe на экране появится такая картинка

[image: image1.png]s Tnasroe Hemn
FRk-1ma

LOATE |, 2 %k Mz gaitna

PRINT "lickeASIC”

ek \ TERCT NPOrpaMME

Ommo penartopa

Toocs mpoxpyTRs

" =
Tedste =
HemenneRHoe BHOMHEHME HABPERHOJ ROMAHNH





   Основное поле (как правило, оно голубого цвета) называется окном редактора, в этом окне мы будем набирать тексты наших программ. В верхней строке экрана расположены слова, образующие главное меню Quick Basic. (Может быть, что этих слов меньше, чем на рисунке. Это значит, что Quick Basic работает в режиме сокращенного, а не полного, меню. Переход в режим полного меню – Full Menus - “спрятан” в пункте Options главного меню.) 

   В середине следующей строки расположено имя программы (точнее, имя файла с этой программой). Пока мы не дали программе другое имя, она будет называться “Untitle”, что значит “не имеющая заголовка”.

   Познакомимся теперь с пунктом File главного меню. Нажмем клавишу Alt - при этом слово File станет светлым на черном фоне. Если теперь нажать Enter, то из этого слова “выпадет” табличка с новыми пунктами - маленькая, если Quick Basic находится в режиме сокращенного меню и большая, если установлен режим полного меню.

New Program

New Program

Open Program

Open Program

Save As...

Merge...

Print...

Save

Exit

Save As...

Save All

Create File...

Load File...

Unload File...

Print...

DOS Shell

Exit

 

   Для первых занятий достаточен режим сокращенного меню, но постепенно мы будем знакомиться и с пунктами расширенного меню.

   New Program - перейти к набору новой программы. Этот пункт следует выполнить, если мы перед тем работали с другой программой и окно редактора не пусто. Quick Basic автоматически выполняет этот пункт при обычном запуске.

   Open Program... - загрузить готовую программу. Quick Basic запросит имя программы в специальном появившемся окне. Кроме имени программы возможно придется указать диск и каталог, в котором она находится. 

   Save As... - записать программу на диск с тем же или новым именем.

   Print... - напечатать текст программы на принтере.

   Exit - выйти из среды программирования Quick Basic.

 

3. Простейшая программа: вывод текста

   Наберем в окне редактора простенькую программу и поясним, что делает каждый из написанных операторов:

'Простейшая программа на Бейсике
- комментарий
CLS
- очистка экрана
FOR i=1 TO 20
- заголовок цикла
   LOCATE i, i
- установить курсор
   PRINT " РЕКЛАМА "
- вывести на экран
NEXT i
- конец цикла
   LOCATE номер_строки, номер_позиции - устанавливает курсор в заданное место экрана. Используется только в текстовом режиме. 
   Номер строки - число от 1 до 25. 
   Номер позиции - число от 1 до 80.

   Редактор в Quick Basic особенный - “думающий”. После того как набрана строка и нажата клавиша ввода, выполняется проверка правильности написанного. При наличии ошибки тут же выдается предупреждение. Чаще всего оно имеет вид “Expecting символ or символ”, т.е. одновременно сообщается, что, по мнению редактора, следует вставить в текст для исправления ошибки.

   Если при наборе строки операторы писать маленькими, строчными буквами, то после нажатия на Enter все они (если в строке нет ошибок!) запишутся большими, прописными буквами. Этот прием помогает заметить ошибку в имени оператора даже при отсутствии предупреждения Quick Basic’а.

   Для сохранения программы на диске выйдем в главное меню (нажмем и отпустим клавишу Alt), раскроем пункт File и выберем Save As... Многоточие обозначает, что при выборе такого пункта Quick Basic вступает в диалог с пользователем. В данном случае на экране появится такое окно:

   Не следует записывать свои программы в каталог Quick Basic’а. Предположим, что они хранятся в каталоге E:\MYBASIC. Если при запуске qb.exe активным был именно этот каталог, то все пункты меню, работающие с диском, будут использовать его по умолчанию.

   Наберем теперь в окошке с именем File Name имя программы - PRG.BAS. (Расширение .BAS можно было и не набирать - Quick Basic добавляет его автоматически.) Но прежде, чем нажать Enter, давайте посмотрим, в каком виде текст программы будет записан на диск. Если в окошке Format точка в круглых скобках стоит возле слова Text, то на диске окажется обычный текст, с которым могут работать и другие программы (например, этот файл можно будет посмотреть и в NC). Если же точка стоит возле слова QuickBASIC, то записанный на диск текст сможет прочитать и понять только сам Quick Basic! Заметим, кстати, что некоторые операции в Quick Basic выполняются только с программой, записанной в текстовом формате.

   Чтобы попасть (при необходимости) в окошко Format, нажмем несколько раз клавишу Tab. Переставим точку с помощью стрелки “вниз” и нажмем Enter. Все - программа записана на диск.

   Для выполнения набранной программы снова выйдем в главное меню и раскроем пункт Run. Выберем в выпавшем окне пункт Start Shift+F5 и нажмем клавишу ввода. Сочетание Shift+F5, написанное возле слова Start - это обозначение так называемой “горячей клавиши”. Нажатие такой комбинации клавиш выполняет те же действия, что и соответствующий пункт меню, но “напрямую” - непосредственно из окна редактора программы, без захода в главное меню.

   В результате выполнения программы PRG.BAS на экране появится “лесенка” из слова “РЕКЛАМА”:

   Задание. Выведите слова “в столбик” (измените оператор LOCATE). Что произойдет, если попытаться вывести слова “в строку”? Объясните, почему.

   Измените текст, выводимый оператором PRINT.

 

4. Возможности редактора cреды программирования Quick Basic

   Каждый редактор текста обязан уметь работать не только с отдельными символами, но и с более крупными кусками - фрагментами текста. Фрагмент может занимать часть одной строки или охватывать одну или несколько строк. Для выделения части строки следует поставить курсор на начало фрагмента и несколько раз нажать комбинацию клавиш Shift+[ ]. Можно выделить текст влево от курсора, если несколько раз нажать на Shift+[ ]. Аналогично выделяются и строки, только нужно воспользоваться комбинацией клавиш Shift+[ ] или Shift+[].

   Работа с фрагментами сосредоточена в пункте Edit главного меню (в скобках указаны “горячие” клавиши):

   Cut (Shift+Del) - удалить выделенный фрагмент. Удаленный фрагмент исчезает с экрана, но временно сохраняется в специальной области памяти. Эту область программисты называют буфером или, по-простому, “карманом”. В буфере хранится только последний удаленный фрагмент.

   Copy (Ctrl+Ins) - выделенный фрагмент копируется в буфер, удаления с экрана не происходит.

   Paste (Shift+Ins) - содержимое буфера вставляется в текст программы (текст “раздвигается”). Место вставки указывает курсор.

   Clear (Del) - выделенный фрагмент удаляется без сохранения в буфере.

С помощью пунктов Cut и Paste можно перенести любой фрагмент в другое место программы; Copy и Paste позволяют “размножить” один и тот же фрагмент. Если нужный фрагмент занимает ровно одну строку, то ее можно не выделять. Строка удаляется комбинацией Ctrl+Y, но при этом, в отличие от других текстовых редакторов, попадает в буфер и может быть восстановлена в том же или другом месте!

   Задание. Используя копирование и вставку фрагментов, создайте программу, которая выведет на экран несколько полосок слов.


5. Работа с монитором в текстовом режиме

	   Монитор современного компьютера может работать в двух режимах: текстовом и графическом. В языке программирования Quick Basic предусмотрены способы вывода информации на экран в обоих режимах.

   Рассмотрим, что и каким образом можно выводить на экран в текстовом режиме.

   По умолчанию Quick Basic работает с монитором в текстовом режиме. В этом режиме на экран можно вывести все символы, которые есть на клавиатуре, и символы псевдографики. В текстовом режиме экран имеет 25 строк, в каждой строке 80 позиций (однако, с помощью оператора WIDTH 40 можно получить 40 “широких” позиций) для вывода символом, следовательно, максимальное количество символов, которое можно разместить на экране в текстовом режиме, равно 2000=25х80.

5.1.Управление цветом в текстовом режиме

   Оператор COLOR цвет_символа, цвет_фона устанавливает цвет выводимого символа и цвет фона под символом.

   Цвет символа - это число от 0 до 15, т.е. символ может быть выведен на экран одним из 16 цветов.

   Цвет фона - это число от 0 до 7.

   Если оператора COLOR в программе нет совсем, то Quick Basic по умолчанию устанавливает цвет 7 (белый) для символа и цвет 0 (черный) для фона.

   Каждый цвет - это комбинация трех основных цветов - красного, зеленого и синего. Каждый цвет может выводиться с низкой или высокой интенсивностью (яркостью). 

Красный

Зеленый

Синий

Обычный

Яркий

0

0

0

 

0 - черный
 

8 – серый
0

0

1

 

1 - синий
 

9 - ярко-синий
0

1

0

 

2 - зеленый
 

10 - ярко-зеленый
0

1

1

 

3 - голубой (точнее, циан)
 

11 - ярко-голубой
1

0

0

 

4 - красный
 

12 - ярко-красный
1

0

1

 

5 - фиолетовый
 

13 - ярко-фиолетовый
1

1

0

 

6 - коричневый
 

14 - желтый
1

1

1

 

7 - белый
 

15 - ярко-белый
   Составим таблицу цветов. Наличие в цвете какой-либо цветовой компоненты будем отмечать единицей, а ее отсутствие - нулем.

   Если к цветам, указанным в таблице, прибавить 16, то символы будут выводиться тем же цветом, но кроме того начнут мигать. Таким образом, если в качестве цвета символа указан код со значением с 16 по 31, то мы получим соответствующий мигающий цвет.

   Пример: обычный красный - 4; мигающий - 20. 

   Задание. Загрузите программу PRG.BAS и сделайте полоски разных цветов. 

 

5.2. Псевдографика

   Некоторые символы не так-то просто вывести на экран просто потому, что их нет на клавиатуре. Это, в частности, относится к символам псевдографики.

   Псевдографика - это символы, с помощью которых можно рисовать линии, рамки и другие графические изображения в текстовом режиме. Символы псевдографики отличаются от обычных символов тем, что занимают всю ширину (высоту) места, отведенного символу на экране. Поэтому они, будучи поставлены рядом, сливаются и образуют непрерывные линии.

   Чтобы вывести на экран символ, отсутствующий на клавиатуре, нужно нажать клавишу Alt и, не отпуская ее, другой рукой набрать код символа (число от 0 до 255). (Некоторые драйверы клавиатуры имеют режим набора псевдосимволов.)

   Приведем коды некоторых символов.

Одинарная рамка:

Двойная рамка:

   Задача. Вывести на экран таблицу символов.

   Решение. С помощью функции CHR$ получим символ по его коду:

символ = CHR$(число) 

и сразу выведем его на экран оператором PRINT. Чтобы все символы поместились на экране, расположим их в виде таблицы 16х16. Пронумеруем строки и столбцы таблицы числами от 0 до 15. Тогда код символа, стоящего на пересечении i-ой строки и j-го столбца, равен значению 16*i + j. Организуем вывод таблицы по строчкам:

CLS

FOR i = 0 TO 15

‘Выводим строку таблицы

   FOR j = 0 TO 15

      kod = 16*i + j

      LOCATE i + 5, 2* j + 5

      PRINT CHR$(kod)

   NEXT j

NEXT i

   Запишем программу под именем TABL.BAS и выполним ее. На экране появится таблица, однако некоторые места в ней останутся пустыми - для этих кодов нет графического представления. Но это не значит, что ими нельзя воспользоваться в программе! Так, например, при выполнении оператора PRINT CHR$(7) раздается звуковой сигнал, PRINT CHR$(12) очищает экран, PRINT CHR$(13) переводит курсор на новую строку и т.д.

   Конечно, чаще всего, те же действия можно выполнить “стандартными” операторами: BEEP - звуковой сигнал, CLS - очистка экрана, перевод строки делается автоматически после вывода на экран (если только в конце оператора PRINT не стоит точка с запятой для подавления перевода строки!). Иногда знание кода символа необходимо. Невозможно, например, вывести оператором PRINT текст, содержащий кавычки, так как последние воспринимаются как ограничители. Чтобы вывести на экран фразу: “Смотрите фильмы в кинотеатре “Космос”!” следует написать

PRINT “Смотрите фильмы в кинотеатре”; CHR$(34);“Космос”; CHR$(34); “!”

или

PRINT “Смотрите фильмы в кинотеатре”+ CHR$(34) +“Космос”+ CHR$(34) + “!”

   Вернемся к программе TABL.BAS. Поскольку вывод символа с кодом 12 эквивалентен оператору CLS, то с экрана сотрется изображения предыдущих символов, и мы их не увидим (разве что заметим, как что-то мелькнуло на экране). Вставим, поэтому, проверку условия: код символа не равен 12 (Так как знака “не равен” нет на клавиатуре, то он изображается двумя символами: <> или ><).

CLS

FOR i = 0 TO 15

   ‘Выводим строку таблицы

   FOR j = 0 TO 15

      kod = 16*i + j

      IF kod <> 12 THEN

         LOCATE i + 5, 2* j + 5

         PRINT CHR$(kod)

      END IF

   NEXT j

NEXT i

   Задания. Пользуясь символами псевдографики “нарисуйте” на экране рамку, лестницу, флаг, забор и т.д.

 

6. Работа с монитором в графическом режиме

   В графическом режиме минимальным элементом, который можно вывести на экран, является не символ, а пиксель. Пиксель (точка экрана) представляет собой крошечный прямоугольник, размеры которого зависят от установленного графического режима. Верхняя левая точка экрана имеет координату (0,0), ось ОХ направлена вправо, ось ОY – вниз, т.е. чем ниже на экране расположена точка, тем больше ее координата по оси OY..

   Оператор SCREEN номер режима переводит режим работы экрана из текстового в графический с заданным номером. Если оператор SCREEN не указан, то подразумевается текстовый режим. Текстовый режим имеет номер 0, поэтому, если нужно из графического режима перейти в текстовый, следует использовать SCREEN 0. Иногда при возврате из графического режима (например, из SCREEN 7) символы текста становятся широкими. Вернуть им нормальные размеры можно только оператором WIDTH 80.Графические режимы различаются разрешением (количеством “точек”) экрана и количеством возможных цветов.

   Например, SCREEN 7 устанавливает разрешение экрана в 320 точек по горизонтали и 200 точек по вертикали. Положение произвольной точки на экране задается двумя числами (x, y) в экранной системе координат. От декартовых координат эта система отличается тем, что ось Oy направлена вниз, а не вверх.

В графическом режиме на экран кроме символов можно выводить и геометрические фигуры: точки, линии, прямоугольники, окружности и т.д. Замкнутые фигуры можно закрашивать.

6.1. Рисование точек и линий

   Оператор PSET (x, y)[, цвет] закрашивает точку экрана заданным цветом.

   Оператор LINE (x1, y1)-(x2, y2)[, цвет] рисует отрезок, соединяющий на экране точки с координатами (x1, y1) и (x2, y2) 

   Оператор LINE - (x2, y2)[, цвет] рисует отрезок, соединяющий последнюю выведенную на экран точку с точкой (x2, y2).

Цвет в этих операторах задается числом от 0 до 15. Если цвет в графических операторах не указан, то все линии рисуются белым цветом. (О том, какие могут быть цвета в различных графических режимах, мы поговорим позже.)

   Пример. Соединить точки (20, 45) и (38, 144).

 

Первый способ: 
LINE (20, 45)-(38, 144)
 

Второй способ: 
PSET (20, 45)
 

 

LINE -(38, 144)

   Задание. Построить на экране треугольник, четырехугольник, пятиугольник; нарисовать домик. Записать программу “Домик” в файл HOUSE.BAS.

 

6.2. Прямоугольники и квадраты

   Рисовать домик отдельными линиями - не самый рациональный способ. Легко увидеть, что домик представляет собой прямоугольник с крышей из двух отрезков. (Так рисуют домик почти все, кроме гениальных фантазеров и профессиональных художников.) А нарисовать прямоугольник можно одним оператором

LINE (x1, y1)-(x2, y2), [цвет], B

   Этот оператор рисует рамку-прямоугольник со сторонами, параллельными границам экрана. Точки (x1, y1) и (x2, y2) суть две вершины прямоугольника, расположенные по одной из его диагоналей. Если параметр цвет отсутствует, то будет нарисована белая рамка. (Но “лишняя” запятая должна остаться!)

   Поскольку точки экрана представляют собой маленькие прямоугольнички, то попытка нарисовать квадрат оператором 

LINE (20, 20)-(120, 120), , B 

потерпит неудачу. Высота этого “квадрата” окажется больше, чем ширина, хотя обе стороны содержат одинаковое число пикселей. Уменьшим число точек в высоте “квадрата” в 1.2 раза (для SCREEN 1!) или во столько же раз увеличим ширину “квадрата” - только тогда мы получим истинный квадрат. Поправочный коэффициент назовем “коэффициентом вытянутости” пикселя.

   Оператор LINE (x1, y1)-(x2, y2), [цвет], BF рисует такой же, но закрашенный прямоугольник.

   Задача. Нарисовать на экране пирамиду из 10 ступенек-прямоугольников.

   Решение. Воспользуемся тем, что пирамида имеет осевую симметрию. Расположим ось пирамиды посредине экрана. Пронумеруем прямоугольники сверху вниз числами от 1 до 10. Очевидно, что значения x1, y1, x2, y2 зависят от номера прямоугольника. В силу симметрии, чтобы получить x1, нужно от 160 вычесть что-то, а чтобы получить x2 - прибавить столько же.

   Составим вспомогательную табличку.

Номер
ступеньки


x1


x2

1

160 - 10

160 + 10

2

160 - 20

160 + 20

3

160 - 30

160 + 30

...

...

...

10

160 - 100

160 + 100

   Легко заметить, что номер ступеньки-прямоугольника и величина добавки связаны зависимостью

добавка = 10*номер.

   Если обозначить через k номер ступеньки, получим формулы для вычисления x1 и x2:

x1 = 160 - 10*k

x2 = 160 + 10*k

   Значения y1 и y2 также зависят от номера и высоты ступеньки h, где h - некоторая заранее заданная величина (например, h = 10):

y1 = 40 + h*k

y2 = y1 + h

   Теперь можно составить программу.

SCREEN 7

h = 10

FOR k = 1 TO 10

   x1 = 160 - 10*k

   x2 = 160 + 10*k

   y1 = 40 + h*k

   y2 = y1 + h

   LINE (x1, y1)-( x2, y2), , B

NEXT k

   Задание. Запишите программу под именем PIRAMIDA.BAS. Измените высоту, ширину, цвет ступенек. Составьте пирамиду из закрашенных прямоугольников; из закрашенных прямоугольников с рамкой другого цвета.

   Заметим, что использование переменной h для обозначения высоты ступенек позволяет проще изменять высоту ступеньки.

6.3. Окружности и дуги

   Оператор CIRCLE (x, y), радиус[, цвет] рисует окружность заданного радиуса и цвета с центром в точке (x, y). Если вся окружность на экране не поместится, то будет нарисована только ее часть. Радиус окружности задается в пикселях, но такое количество пикселей укладывается только по горизонтали. Размер вертикального радиуса (в пикселях) зависит от номера графического режима.

   Задача. Нарисовать несколько концентрических окружностей (“мишень”).

Решение.

SCREEN 7

FOR r = 5 TO 65 STEP 5

   CIRCLE (160, 100), r, 3

NEXT r

   Задание. Запишите программу под именем MISHEN.BAS. Сделайте шаг в цикле равным 3, 2, затем 1 (но не 0, так как программа “зациклится”!). Объясните возникающий при этом эффект. Измените положение “мишени” на экране так, чтобы ее часть ушла за экран. Введите зависимость положения центра окружности от ее радиуса. Поэкспериментируйте с несколькими разноцветными “мишенями”.

   Тем же оператором можно нарисовать и дугу, только дополнительно потребуется задать значения начального и конечного углов дуги:

CIRCLE (x, y), радиус, [цвет], нач_угол, кон_угол

   Если поместить в точку (x, y) начало математической системы координат, то начальный и конечный углы откладываются от оси Ox в направлении против часовой стрелки.

   Начальный угол может быть больше конечного, но принцип рисования дуги останется прежним.

   Углы дуги должны задаваться в радианах. Поскольку это не всегда удобно, автор рекомендует переводить градусы в радианы каждый раз, как это потребуется. Коэффициент пропорциональности QR вычисляется по формуле QR = /180, где всем известная константа. Конечно, можно использовать и конкретное значение коэффициента: QR  0.0174329252.

   Пример. Изобразите на листе бумаги, что получится на экране после выполнения программы:

SCREEN 7

QR = 0.0174329252

CIRCLE (100, 50), 60, 2, 270*QR, 0

CIRCLE (220, 50), 60, 2, 180*QR, 270*QR

CIRCLE (100, 150), 60, 2, 0, 90*QR

CIRCLE (220, 150), 60, 2, 90*QR, 180*QR

   Правильность ответа проверьте на компьютере.

   Замечание. Математики рассматривают как положительные, так и отрицательные углы. В операторе CIRCLE также могут использоваться отрицательные значения для начального и конечного углов, однако здесь знак минус “сообщает”, что следует не только нарисовать дугу, но и провести один или оба радиуса. Измените предыдущую программу, посмотрите, что получится на экране. Учтите, что с точки зрения Quick Basic’а “отрицательный” нуль ничем не отличается от обычного, поэтому замените его очень маленьким отрицательным числом, например, - 0.001.

 

   Задания для самостоятельной работы

Составьте программы, которые выводят на экран следующие картинки:

6.4. Графические режимы

   SCREEN 1 - это устаревший графический режим, используемый в Quick Basic только для совместимости с некоторыми мониторами. В этом режиме доступны одновременно только 4 цвета (включая цвет фона). Одному и тому же номеру соответствуют разные цвета в зависимости от выбранной палитры:

Номер цвета

Палитра №1

Палитра №2

1

голубой (3)

зеленый (2)

2

фиолетовый (5)

красный (4)

3

белый (7)

коричневый (6)

   Оператор COLOR в этом режиме имеет свой особый синтаксис:

COLOR цвет_фона, номер_палитры

   Выбор цвета фона намного богаче - он может быть любым из 16 основных цветов (от 0 до 15).

   По умолчанию SCREEN 1 устанавливает первую палитру.

   SCREEN 2 задает разрешение 640 точек по горизонтали на 200 точек по вертикали. В этом графическом режиме возможны только два цвета - черный и белый, заданные “раз и навсегда”. Использование оператора COLOR в этом режиме некорректно и вызовет сообщение об ошибке “Illegal function call” (“Недопустимый вызов функции”).

   Высота пикселя в этом режиме больше его ширины в 2.4 раза.

SCREEN 7 задает разрешение 320 точек по горизонтали на 200 точек по вертикали, следовательно, “коэффициент вытянутости” пикселя, как и в SCREEN 1 равен 1.2..

   В этом и следующих режимах оператор COLOR управляет цветом линий, цветом фона экрана и цветом символов текста. Можно указывать цвет рисования и в каждом графическом операторе (если цвет пропущен, то линия рисуется цветом, заданным в операторе COLOR, а если нет и оператора - то ярко-белым цветом). Оператор COLOR имеет следующий вид:

COLOR цвет_линии, цвет_фона

   Цвет линии - число от 0 до 15, цвет фона - число от 0 до 7. Заданные параметры распространяются и на текстовый вывод, но в отличие от текстового режима нельзя вывести на экран мигающие символы. Кроме того, изменение цвета фона действует и на уже выведенный текст!

   SCREEN 8 задает разрешение 640 точек по горизонтали на 200 точек по вертикали (“коэффициент вытянутости” = 2.4), а SCREEN 9 - 640 точек по горизонтали на 350 точек по вертикали. Точное значение “коэффициента вытянутости“ в SCREEN 9 равно 48/35. что практически равно 1.37.

   Для мониторов типа VGA и SVGA возможны и другие графические режимы. Чаще всего используются режим SCREEN 12 с разрешением 640 точек по горизонтали на 480 точек по вертикали, в котором “коэффициент вытянутости” в точности равен 1. Однако в этом режиме в операторе COLOR указывается только один параметр - цвет линии (символа). Цвет фона оператором не задается!

   Задание. Загрузите программу PIRAMIDA.BAS и посмотрите результат выполнения в рассмотренных графических режимах. Сделайте ступеньки пирамиды разных цветов. Подпишите рисунок (используйте оператор COLOR для задания цвета фона и символов текста).

 

6.5. Закраска произвольной области

   Если на экране нарисована замкнутая кривая, то область внутри нее можно закрасить. Также можно закрасить и область, оставшуюся снаружи от кривой. Для этого используется оператор

   PAINT (x, y), цвет_закраски, цвет_границы

   Точка (x, y) должна попасть внутрь области, подлежащей закрашиванию.

   Если граница области “дырявая”, то краска выльется за пределы области, даже если дырка всего в один пиксель. То же самое произойдет, если граница области содержит по недосмотру кусочек другого цвета - для оператора PAINT это все равно что дырка.

   Пример. Нарисовать флаг Японии - красный круг на белом фоне.

‘--------- Флаг Японии ---------

SCREEN 7

LINE (10, 10)-(100, 50), 15, BF

CIRCLE (55, 30), 20, 12

PAINT (55, 30), 12, 12

   Задание. Нарисовать флаги предложенных ниже стран или любой другой страны.

	   Чтобы упростить построение сложного контура, можно составить его из нескольких простых. Например, крест на флаге Швеции сделать из двух закрашенных прямоугольников, серп на флаге Турции - из двух кругов, один из которых закрасить белым, а другой, меньший, красным. 

   При закраске окружности в качестве начальной точки для оператора PAINT удобно брать центр окружности, в других случаях, особенно при закраске маленьких областей, легко “промахнуться” мимо области. Чтобы проконтролировать расположение начальной точки, заменим оператор PAINT оператором PSET(x, y). Убедимся в том, что точка (x, y) попала в нужное место, и после этого вернем оператор PAINT на свое законное место.


7. Случайные числа

	   Случайное число - это число, величина которого не может быть заранее известна. Классический пример - число очков, выпадающих при бросании игрального кубика.

   В Quick Basic имеется функция RND, которая возвращает случайное значение из интервала (0, 1). Познакомимся с ней поближе. Составим программу вывода первых 5 случайных чисел на экран.

Программа

Результат

'Получение случайных чисел

FOR i = 1 TO 5

   x = RND

   PRINT x

NEXT i

         .7055475

         .533424

         .5795186

         .301948

         .7747401

   Если выполнить эту программу несколько раз, то мы с удивлением обнаружим, что на экран выдается одна и та же последовательность “случайных” чисел! Это происходит потому, что на самом деле каждое следующее случайное число получается из предыдущего по хитрой формуле, которая имитирует отсутствие закономерности в последовательности чисел. А поскольку первое число последовательности одно и то же, то повторяться будут и остальные числа.

   Чтобы получать разные последовательности нужно при каждом запуске брать новую “затравку”. Для изменения “затравки” используем команду RANDOMIZE TIMER.

   RANDOMIZE получает от функции TIMER значение системного времени и формирует “затравку” для RND. При следующем запуске программы функция TIMER вернет другое значение, и получится другая последовательность чисел.

   Задача. Получить случайное целое число x в пределах от 1 до N

   Решение. Умножим неравенство 0 < RND < 1 на N > 0. При умножении на положительное число знак неравенства сохраняется, поэтому 0 < N*RND < N. Чтобы получить целую часть результата воспользуемся функцией INT . Но INT(N*RND) изменяется в пределах от 0 до N-1. Осталось только прибавить единицу, чтобы получить окончательную формулу:

x = INT(N*RND) + 1

   Задача. “Зажечь” на экране (в текстовом режиме) 100 звездочек.

   Решение. Для каждой звездочки выберем для вывода случайным образом номера строки и позиции. Учитывая, что строк на текстовом экране 25, а позиций 80, получим программу

RANDOMIZE TIMER

FOR i = 1 TO 100

   nstr = INT(25*RND) + 1

   npos = INT(80*RND) + 1

   LOCATE nstr, npos: PRINT “*”;

NEXT i

   Точка с запятой после оператора PRINT нужна, чтобы отменить перевод строки после вывода звездочки, иначе при выводе в последнюю строку экрана вся картинка сместится вверх.

   Задание. Изменить программу так, чтобы она выводила на экран в графическом режиме 1000 точек случайного цвета. Замените точки окружностями или прямоугольниками.

8. Музыкальные и звуковые возможности

   Quick Basic использует известные музыкантам буквенные обозначения нот. Для повышения или понижения основной ноты на полтона используются знаки альтерации - диез (# или +) или бемоль (-), которые ставятся после ноты.

   Длительность нот обозначаются числами: 1 - целая, 2 - половинная, 4 - четвертная, 8 - восьмая, 16 - шестнадцатая. Число может стоять после каждой ноты (c4, e-2, a8), либо, если длительность нескольких подряд идущих нот одинакова, указывается один раз после английской буквы L (L1, L2, L4 и т.д.). Аналогично указывается длительность пауз - p1, p2, p4 и т.д.

   Октавы обозначаются английской буквой o с указанием номера от 0 до 6, причем первой октаве соответствует номер 3. Кроме непосредственного указания октавы, можно использовать символы > (переход на октаву вверх) и < (переход на октаву вниз).

   Мелодия, взятая как и любая текстовая строка, в кавычки, исполняется оператором PLAY. Длинную мелодию лучше записать в нескольких операторах, но так, чтобы Quick Basic смог понять каждый их них.

   Буквы в записи мелодии могут быть как строчными, так и прописными. Можно, например, использовать прописные буквы для установки длительностей, октав, режимов и т.д., а строчные - для самих нот.

   Пример. Наберите и выполните следующую программу:

   PLAY “O2 L8 ececfe d4 < ggg a16 b16 > cc c4”

   Обратите внимание на запись длительностей нот. Указание L8 действует на все ноты, не имеющие “своей” длительности. С другой стороны, длительность, приписанная конкретной ноте, не распространяется на другие ноты.

   Мелодия “Чижика” исполнялась в нормальном стиле с очень небольшими паузами между нотами. Если в начале мелодии поставить ML, то этих пауз не будет совсем. Такой стиль музыканты называют “легато” (legato). Противоположным является стиль “стаккато” (staccato) - отрывистое, с большими паузами между нотами исполнение мелодии. Этот стиль задается сочетанием MS. И, наконец, чтобы после ML или MS вернуться к нормальному стилю, следует поставить в нужное место MN.

   Темп исполнения (не путайте с разной длительностью нот!) может изменяться в широких пределах. Самый медленный обозначается T32, а самый быстрый - T255. По умолчанию устанавливается темп T120 - 120 четвертей в минуту.

   Задание. Измените стиль и темп исполнения написанной музыкальной фразы. Продолжите мелодию “Чижика”. Подберите мелодию какой-либо другой известной песни.

   Напоследок упомянем о двух режимах исполнения мелодии. В режиме MF (устанавливается по умолчанию) программа ждет завершения оператора PLAY. В режиме MB программа продолжает выполняться на фоне музыки.

 


9. Ввод с клавиатуры

	   Оператор INPUT список_переменных принимает числовые или текстовые значения с клавиатуры и присваивает их переменным из списка. Программа, дойдя до оператора INPUT, выводит на дисплей знак вопроса и ждет ввода с клавиатуры. Для общения с клавиатурой используется небольшая область памяти (буфер клавиатуры). При нажатии клавиши соответствующий символ (точнее, его код) попадает в буфер клавиатуры, оператор INPUT забирает его оттуда и отображает на экране. Ввод заканчивается при нажатии клавиши Enter.

   Имена переменных, предназначенных для хранения текстовых значений, должны заканчиваться знаком $. Имена переменных, хранящих числовые значения, могут в зависимости от типа значения заканчиваться знаком !, # или &. При отсутствии этих знаков в имени переменной считается, что она имеет тип, соответствующий восклицательному знаку. (Подробнее о типах данных рассказано в соответствующем разделе.)

   Примеры:

А)   В программе   
INPUT x, y
 

PRINT x, y
 

 

На дисплее   
? 30, 47 
 

30   47
   В результате переменная “x” получит значение 30, а переменная “y” - значение 47.

   Если повторно запустить программу и ввести только одно значение, например, 30 , то Quick Basic предложит повторить ввод - выдаст сообщение “Redo from start”. То же самое сообщение мы получим, если захотим ввести значений больше, чем переменных в списке оператора INPUT.

Б)   В программе   
INPUT word$
 

PRINT word$
 

 

На дисплее   
? Антон 
 

Антон
   В результате переменная word$ будет содержать слово “Антон”. Обратите внимание, что вводимое слово в кавычки не берется, хотя в операторе присваивания word$ = “Антон” они обязательны!

   Запустим программу повторно и наберем на клавиатуре слово “Сергей”. К нашему удивлению, буква “р” не появляется на экране, хотя мы изо всех сил жмем ее на клавиатуре. Увы! Стандартными средствами языка ввести с клавиатуры строчную русскую “р” невозможно. (Мы вернемся к этой проблеме в разделе “Дополнительные возможности”.)

   Можно ли ввести число в переменную word$? Давайте в ответ на приглашение введем число 12. Программа “ответит” тем же значением, т.е. переменная word$ содержит сейчас текстовое (но не числовое!) значение “12”. Операции над текстовыми значениями (строками символов) выполняются совсем по другому, чем над числовыми. Выполним, например, программу:

n = 12 + 3

n$ = “12” + “3”

PRINT n, n$

   На дисплей будут выведены два значения: 15 и 123. Но если первое значение - число, полученное в результате арифметической операции сложения, то второе - текстовая константа “123”, которая получена в результата особой операции “склеивания” двух строк.

   Из строки символов можно выделить любую ее часть с помощью функции MID$:

   часть = MID$(строка, начальная_позиция, количество_символов)

   Оператор INPUT$(число) вводит заданное число символов без их отображения на экране. Нажимать Enter после ввода не требуется.

   Этот оператор удобно использовать при вводе пароля. Пример:

‘Запрос пароля
password$ = INPUT$(4)

IF password$ <> “aq” + CHR$(8) + CHR$(27)  THEN STOP

‘Основная программа

   ...

   Чтобы запустить основную программу, нужно нажать две буквенных клавиши - “a” и ”q”. затем клавиши Backspace и Esc, коды которых (8 и 27) преобразуются функцией CHR$ в символы.

   С помощью операторов INPUT и INPUT$ нельзя обработать нажатие ряда клавиш: функциональных, клавиш управления курсором, Home, End, PageUp, PageDn и некоторых других. Для этого используется функция INKEY$. Эта функция берет символы из буфера клавиатуры (если он не пуст!), но не ждет нажатия клавиши. Если буфер пуст, INKEY$ возвращает значение “пустая строка” - две кавычки, между которыми ничего нет.

   При нажатии клавиши в буфер клавиатуры помещается одно или два числа - код клавиши. Если код клавиши состоит из двух чисел, то первое число обязательно равно 0. Коды некоторых клавиш приведены в таблице.

Клавиша

BS
Enter
Esc
пробел
+
2
G
g
я
Код

8

13

27

32

43

50

71

103

239

Клавиша

F1
F2
Home

PageUp



Ins
Код
0; 59

0; 60

0; 71

0; 72

0; 73

0; 75

0; 77

0; 80

0; 82

   Функция INKEY$ возвращает символ, соответствующий коду в буфере клавиатуры. Если она обнаруживает 0 в буфере клавиатуры, то возвращается “склейка” двух символов - CHR$(0) и следующего за ним.

Клавиши Alt, Shift и Ctrl являются клавишами-модификаторами: они изменяют код нажатой одновременно с ними клавиши, но сами не имеют кода. Например, клавиша F2, нажатая с клавишей Shift, генерирует код 0; 85, с клавишей Ctrl - 0; 95, с клавишей Alt - 0; 105.

   Задача. Написать программу, которая выводит на экран код нажатой клавиши. Выход из программы - по клавише Esc.

Решение. Для ввода символа воспользуемся уже известной нам функцией INKEY$. а для перевода символа в код - функцией ASC:

код = ASC(символ).

   Запишем алгоритм работы программы:

1. Ждем нажатия клавиши. 

2. Выделяем первый символ полученной (из буфера клавиатуры) строки и выводим его код на печать. 

3. Если первый символ строки равен CHR$(27), то заканчиваем работу. 

4. Если первый символ строки равен CHR$(0), то выделяем и печатаем код второго символа. 

5. Переходим к пункту 1. 

‘------------------Программа ------------------

‘Ждем нажатия клавиши
10 s$ = INKEY$: IF s$ = "" THEN GOTO 10

‘Выделяем и печатаем первый символ
first$ = MID$(s$, 1, 1)

PRINT ASC(first$)

‘Анализируем первый символ
IF first$ = CHR$(27) THEN STOP

‘Выделяем и печатаем второй символ
IF first$ = CHR$(0) THEN

   next$ = MID$(s$, 2, 1)

   PRINT ASC(next$)

END IF

GOTO 10

 

10. Анализ нажатой клавиши

   Во многих программах взаимодействие пользователя с программой происходит с помощью клавиш-стрелок влево, вправо, вверх, вниз. Запишем строки, возвращаемые функцией INKEY$ при нажатии этих клавиш, в переменные LF$ (от английского left - влево), RT$ (от английского right - вправо), UP$ (от английского up - вверх), DN$ (от английского down - вниз).

LF$ = CHR$(0) + CHR$(75)

RT$ = CHR$(0) + CHR$(77)

UP$ = CHR$(0) + CHR$(72)

DN$ = CHR$(0) + CHR$(80)

   Пусть в переменной s$ находится значение, полученное от функции INKEY$. Проанализируем ее содержимое с помощью нескольких IF:

IF s$ = LF$  THEN

‘Операторы, выполняемые по нажатию стрелки влево
   ...

ELSE IF s$ = RT$  THEN

‘Операторы, выполняемые по нажатию стрелки вправо
   ...

ELSE IF s$ = UP$  THEN

‘Операторы, выполняемые по нажатию стрелки вверх
   ...

ELSE IF s$ = DN$  THEN

‘Операторы, выполняемые по нажатию стрелки вниз
   ...

ELSE

‘Была нажата какая-то другая клавиша
   ...

END IF

   Более компактно эти же проверки можно сделать с помощью оператора SELECT CASE (выбрать вариант):

SELECT CASE выражение
- “заголовок” оператора выбора
   CASE значение_1 : действие_1
- выполняется, если выражение = значение_1
   CASE значение_2 : действие_2
- выполняется, если выражение = значение_2
   CASE ELSE : иное_действие
- выполняется, если значение выражения не совпадает ни с одним из перечисленных значений 
END SELECT
- конец оператора выбора
   Задача. При нажатии любой из клавиш-стрелок, напечатать сообщение, какая стрелка нажата, при нажатии на Esc выйти из программы.

   Решение. Напишем сначала алгоритм:

1. Задаем константы. 
Кроме констант LF$, RT$, UP$ и DN$ зададим еще Esc$ = CHR$(27). 

2. Ждем нажатия клавиши. 

3. Анализируем полученную символьную строку и выполняем заданное действие (печать сообщения или останов). 

4. Чтобы не запускать программу каждый раз заново, переходим к пункту 2.

‘---------Программа-----------

‘Задаем константы

LF$ = CHR$(0) + CHR$(75)

RT$ = CHR$(0) + CHR$(77)

UP$ = CHR$(0) + CHR$(72)

DN$ = CHR$(0) + CHR$(80)

Esc$ = CHR$(27)

‘Ждем нажатия клавиши
10 s$ = INKEY$: IF s$ = “” THEN GOTO 10

‘Анализируем строку s$

SELECT CASE s$

   CASE LF$: PRINT “Нажата стрелка влево”

   CASE RT$: PRINT “Нажата стрелка вправо”

   CASE UP$: PRINT “Нажата стрелка вверх”

   CASE DN$: PRINT “Нажата стрелка вниз”

   CASE Esc$: STOP

   CASE ELSE: GOTO 10

END SELECT

GOTO 10

   Последний оператор GOTO 10 “зацикливает” программу, оставляя ей единственный выход - по клавише Esc. При такой организации программы у нас есть возможность проверить реакцию на каждую из клавиш-стрелок, не запуская каждый раз программу заново. Оператор GOTO 10 в CASE ELSE выполняет другую важную функцию: не позволяет программе выйти из оператора SELECT до тех пор, пока не будет нажата одна из нужных клавиш. Нажатия на другие клавиши (кроме аварийного останова по Ctrl+Break) будут игнорироваться.

   Задание. Записать программу под именем ARROW.BAS. Добавить в оператор выбора анализ двух-трех других клавиш (“+”, “-”, “1”, “2”, “F1” и т.д.). 

   Указание. Для клавиш с обычными символами нет необходимости знать их код. Нужно просто указать этот символ в кавычках после CASE.

 

11. Управление окружностью

   Нарисуем на экране некоторую фигуру, например, окружность. Как заставить ее двигаться при нажатии на клавишу-стрелку? Предположим, что мы нажали на стрелку влево. Сотрем на старом месте окружность и нарисуем такую же, но смещенную на несколько пикселей влево. Если была нажата стрелка вправо, то смещенную окружность будем рисовать на несколько пикселей вправо и т.д.

   Для хранения параметров “старой” и “новой” окружности введем переменные:

   x0, y0 - координаты центра “старой” окружности; r0 - ее радиус.

x, y - координаты центра “новой” окружности; r - ее радиус.

   Приведем алгоритм решения.

1. Задаем начальные значения переменных. 

2. Рисуем окружность. 

3. Ждем нажатия клавиши. 

4. Вычисляем координаты центра “новой” окружности. 

5. Стираем “старую” и рисуем “новую” окружность.
(Чтобы стереть окружность, нужно нарисовать ее цветом фона!) 

6. Заменяем параметры “старой” окружности новыми значениями и переходим к пункту 3. 

   В пункте 6 алгоритма происходит обновление переменных x0, y0, r0. Им присваиваются значения x, y, r, и с этого момента окружность становится “старой”. При следующем проходе программа будет оперировать с новыми значениями параметров “старой” окружности.

   (Возможен вариант:

   5’. Стираем “старую” окружность.

   6’. Заменяем координаты центра “старой” окружности новыми значениями и переходим к пункту 2.)

   Программу “Управление окружностью” напишем на основе ARROW.BAS.

‘-------- Программа “Управление окружностью” --------

‘Задаем константы
LF$ = CHR$(0) + CHR$(75)

RT$ = CHR$(0) + CHR$(77)

UP$ = CHR$(0) + CHR$(72)

DN$ = CHR$(0) + CHR$(80)

Esc$ = CHR$(27)

‘Задаем начальные значения переменных
SCREEN 7

x0 = 160: y0 = 100: r0 = 10

‘Рисуем окружность
CIRCLE (x0, y0), r0

‘Ждем нажатия клавиши
10 s$ = INKEY$: IF s$ = “” THEN GOTO 10

‘Анализируем строку s$  и вычисляем параметры “новой” окружности
SELECT CASE s$

   CASE LF$: x = x0 - 5: y = y0: r = r0

   CASE RT$: x = x0 + 5: y = y0: r = r0

   CASE UP$: x = x0: y = y0 - 5: r = r0

   CASE DN$: x = x0: y = y0 + 5: r = r0

   CASE Esc$: STOP

   CASE ELSE: GOTO 10

END SELECT

‘Стираем “старую” и рисуем “новую” окружность
CIRCLE (x0, y0), r0, 0

CIRCLE (x, y), r

‘Заменяем параметры “старой” окружности новыми значениями
x0 = x: y0 = y: r0 = r

GOTO 10

   Запишем программу под именем CIRCLES.BAS, убедимся, что она работает, и попробуем ее улучшить. Во-первых, нерационально повторять в каждом CASE одинаковые присваивания x = x0, y = y0, r = r0. Удобнее сделать все три присваивания перед оператором выбора, а в каждом варианте вычислять только тот параметр, который действительно изменялся. Во-вторых, мы заложили в программу возможность изменения радиуса окружности, но не воспользовались этим. Назначим клавишу “>” для увеличения, а клавишу “<” - для уменьшения радиуса. Наконец, если “стирать” окружность не цветом фона, а любым другим, на экране будет оставаться след - узор из окружностей.

   Задание. Реализуйте все улучшения программы. Придумайте, как изменять цвет оставляемого окружностью следа и шаг перемещения окружности.

 

12. Операторы GET и PUT

   Оператор GET позволяет запомнить часть графического экрана в некотором массиве, а оператор PUT - вывести запомненное изображение на экран. Синтаксис оператора GET:

   GET (x1, y1)-(x2, y2), массив - запомнить прямоугольную область.

   Точки (x1, y1) и (x2, y2) - две вершины прямоугольника, лежащие на одной из его диагоналей. Размеры прямоугольника и его содержимое запоминаются в массиве.

   Синтаксис оператора PUT:

   PUT (x, y), массив [, способ_вывода] - вывести на экран картинку из массива заданным способом.

[image: image16.png]@1, y1)

L
O

(2, y2)

GET
MACCHB

[£3)]

=
O





   Поскольку в массиве хранятся и размеры картинки, в операторе PUT указывается только одна точка - верхний левый угол прямоугольной области для вывода картинки.

 

12.1. Понятие массива

   Проиллюстрируем наглядно понятия простой переменной и массива. Простую переменную можно представить в виде ящичка, на котором написано имя переменной (в нашем случае - A). В ящичек можно положить любое число - значение переменной (в нашем случае - 5).

   Массив можно представить как ящик с перегородками. В каждый из получившихся “отсеков” можно положить свое число. Отсеки пронумерованы, поэтому имя каждого отсека состоит из двух частей: общего имени ящика и номера отсека. Но такие обозначения как A0, A1, A2 используются в математике, в языках программирования элемент массива обозначается A(0), A(1), A(2) и т.д.

   Каждый массив должен быть описан в операторе DIM (от английского dimentions - размеры):

   DIM имя_массива(номер_последнего_элемента)

   Поскольку нумерация элементов массива в Quick Basic начинается с нуля, то количество элементов массива (длина массива) оказывается на единицу больше номера последнего элемента. Например, DIM A(7) описывает массив из восьми элементов. Иногда, для удобства работы, мы будем “забывать” о нулевом элементе и указывать в описании массива его длину.

   Элементы массива должны иметь значения одного и того же типа. Для указания типа элементов к имени массива добавляется знак $, если элементы массива суть строки символов, или знак %, если все элементы массива - целые числа. (Подробнее о типах данных поговорим в соответствующем разделе.)

   Указатели типа являются частью имени как обычной переменной так и массива, поэтому A% и A$ - это имена разных переменных!

   В операторах GET и PUT будем использовать только массивы целых чисел. Размер массива должен быть достаточно большим, чтобы в него поместилась необходимая информация (сохранилась нужная картинка с экрана). В литературе [1] приводятся точные формулы, но можно пользоваться более простой приближенной формулой

размер_массива = 0.3*dx*dy,

где dx = x2 - x1 + 1, dy = y2 - y1 +1 - размеры (в пикселях) прямоугольной области, полностью закрывающую интересующую нас картинку, по горизонтали и вертикали. Если размер массива, вычисленный по приближенной формуле, окажется недостаточным, Quick Basic выдаст предупреждение “Illegal function call” (“Неверный вызов функции”) и остановит выполнение программы на строке с оператором GET. В этом случае следует несколько увеличить размер массива и запустить программу еще раз.

‘---------------- Программа ------------------

‘----“Копирование прямоугольника” ---

SCREEN 7

DIM a%(     )  ‘Размер массива укажем позже

LINE (20, 20)-(79, 69), 5, BF

GET (20, 20)-(79, 69), a%

SLEEP 1

PUT (120, 20), a%

   Программа выводит на экран закрашенный прямоугольник, для которого dx=60, dy=50. Вычисляя по приближенной формуле размер массива, получим значение 900 (точная формула дает значение 802). Укажем теперь размер массива и запустим программу. На экране появится фиолетовый прямоугольник, а ровно через одну секунду (SLEEP n дает задержку в n секунд) рядом появится его дубликат.

   Задание. Запишите программу под именем GETPUT.BAS. Измените положение второго прямоугольника на экране. Что получится, если прямоугольники будут пересекаться?

 

12.2. Способы вывода оператором PUT

   Если в программе “Копирование прямоугольника” в последней строке поставить PUT (30, 30), a% , то на экране появится такая вот картинка.

Мы видим, что в результате наложения двух прямоугольников одного цвета образовалось черное пятно. Это произошло потому, что при выводе картинки оператором PUT над цветом точки экрана и цветом точки картинки, попадающей на это место, производится логическая операция “исключающее или” (краткое обозначение - XOR). Особенность ее состоит в том, что для любого числа n выполняется равенство: n XOR n = 0. Следовательно, наложение точек одинакового цвета даст цвет 0, т.е. цвет фона.

Способ вывода XOR принят в операторе PUT по умолчанию, т.е. если не указан другой способ вывода.Способы вывода определяются последним параметром.

   PSET - вывод картинки без изменения ее цвета. В этом случае никакие операции над цветами картинки и экрана не производятся.

   PRESET - замена цветов картинки на дополнительные. Дополнительный цвет вычисляется по формуле: цвет = 15 - исходный_цвет.

   OR - цвет картинки определяется результатом логической операции “или”.

   AND - цвет картинки определяется результатом логической операции “и”.

   Задание. Замените в программе “Копирование прямоугольника” последнюю строку циклом

FOR x = 30 TO 130 STEP 10

   PUT (x, x), a%

NEXT x

   Нарисуйте границу исходного прямоугольника другим цветом. Измените способ вывода в операторе PUT.

   “Связку” операторов GET и PUT можно использовать для создания периодического узора. Для этого рисуем небольшой прямоугольный фрагмент узора, а затем заполняем фрагментами экран.

‘----------- Программа “Узор” -----------

SCREEN 7

DIM a%(130)

‘Рисуем и запоминаем фрагмент узора
FOR r = 0 TO 10 STEP 2

  LINE (10 - r, 10 - r)- (10 + r, 10 + r), r, B

NEXT r

GET (0, 0)-(20, 20), a%

‘Размножаем фрагмент
CLS

FOR x = 50 TO 250 STEP 20

   FOR y = 40 TO 140 STEP 20

      PUT (x, y), a%, PSET

   NEXT y

NEXT x

 


13. Мультипликация (движение картинки)

	   Управление картинкой на “пустом” экране ничем не отличается от управления окружностью - нарисовал - стер - нарисовал на новом месте. Но если на экране уже есть какое-то изображение, на фоне которого должна двигаться картинка, то возникает проблема - как стереть “старое” изображение, чтобы не испортить фон.

   Здесь нам поможет “связка” операторов GET и PUT.

   Вспомним, что при выводе оператором PUT картинки на экран, по умолчанию над цветом точки экрана и цветом точки картинки выполняется логическая операция “исключающее или”. При этом на месте точек, совпадающих по цвету, появляются черные “дыры” (точнее, прозрачные “дыры”, через которые виден цвет фона). Значит, если наложить картинку саму на себя, то она исчезнет с экрана. Если фон экрана уже содержит изображение, то при первом выводе картинки ее цвета изменятся (к сожалению!), но зато при втором выводе восстановится фоновое изображение. В ряде случаев изменение цветов картинки нежелательно, тогда приходится использовать другой алгоритм, который мы разберем позже.

   Итак, за основу возьмем алгоритм управления окружностью и программу CIRCLES.BAS. Дополним алгоритм “нулевым” пунктом:

      0. Рисуем и запоминаем картинку

и немного модифицируем остальные пункты:

1. Задаем константы. 

2. Задаем исходное положение картинки и выводим ее на экран. 

3. Ждем нажатия клавиши. 

4. Вычисляем новое положение картинки. 

5. Стираем картинку на старом и выводим ее на новом месте. 

6. Обновляем переменные, отвечающие за положение картинки, и переходим к пункту 3. 

   Положение картинки на экране задается следующим образом: фиксируется некоторая точка картинки и задаются ее координаты на экране. За эту точку картинка “привязывается” к экрану, поэтому ее иногда называют точкой привязки. В качестве точки привязки возьмем левый верхний угол (вершину) прямоугольной области, в которой размещена картинка.

   Для рисования картинки можно использовать любые графические операторы, но если картинка маленькая, удобнее задавать ее по точкам. Посмотрим на увеличенный рисунок рожицы.

 

   В таблице справа записаны цвета отдельных точек картинки. (Подразумевается, что пустые клетки таблицы заполнены нулями.)

   В программе на Quick Basic таблицу чисел удобно задать с помощью оператора DATA.

‘Задаем картинку
DATA 0, 0, 7, 7, 7, 7, 7, 0, 0

DATA 0, 8, 0, 0, 0, 0, 0, 8, 0

DATA 8, 0, 0, 8, 0, 8, 0, 0, 8

DATA 8, 0, 3, 3, 0, 3, 3, 0, 8

DATA 8, 0, 0, 0, 0, 0, 0, 0, 8

DATA 8, 0, 0, 12, 12, 12, 0, 0, 8

DATA 0, 8, 0, 0, 0, 0, 0, 8, 0

DATA 0, 0, 8, 8, 8, 8, 8, 0, 0

‘Выводим ее на экран
SCREEN 7

DIM a%(40)

FOR y = 0 TO 7

   FOR x = 0 TO 8

      READ c: PSET (x, y), c

   NEXT x

NEXT y

‘Запоминаем картинку в массиве и очищаем экран
GET (0, 0)-(8, 7), a%

CLS

‘Задаем константы
lf$ = CHR$(0) + CHR$(75)

rt$ = CHR$(0) + CHR$(77)

up$ = CHR$(0) + CHR$(72)

dn$ = CHR$(0) + CHR$(80)

‘Задаем исходное положение картинки и выводим ее на экран
x0 = 165: y0 = 45: x = x0: y = y0

PUT (x0, y0), a%

‘Ждем нажатия клавиши
10 w$ = INKEY$: IF w$ = "" THEN GOTO 10

‘Вычисляем новое положение картинки
SELECT CASE w$

   CASE  lf$: x = x0 - 5

   CASE  rt$: x = x0 + 5

   CASE  up$: y = y0 - 5

   CASE  dn$:  y = y0 + 5

   CASE CHR$(27): STOP

   CASE ELSE: GOTO 10

END SELECT

‘Стираем картинку на старом и выводим на новом месте
PUT (x0, y0), a%

PUT (x, y), a%

‘Обновляем переменные
x0 = x: y0 = y

GOTO 10

   Запишем программу в файл ROZICA.BAS и попробуем ее выполнить. Пока “рожица” бегает по экрану - все хорошо. Но при попытке увести ее за экран появляется сообщение об ошибке “Illegal function call” и высвечивается строка с оператором PUT(x, y), a%. Что ж, будем контролировать значения переменных x и y, а при необходимости, восстанавливать прежние значения. Это просто для случаев движения влево или вверх, так как (x, y) - координаты левого верхнего угла картинки:

CASE lf$: x = x0 - 5: IF x < 0 THEN x = x0

CASE up$: y = y0 - 5: IF y < 0 THEN y = y0

    

Если же мы движемся вправо или вниз, то необходимо чтобы правый нижний угол картинки еще помещался на экране. Чтобы не менять во всей программе параметры экрана при их изменениях, введем переменные xmax - номер самой правой точки экрана и ymax - номер самой нижней точки экрана (для SCREEN 7 xmax = 319, ymax = 199). То же самое относится и к изменению размеров картинки - пусть она при выводе в левый верхний угол экрана занимает место от 0 до dx по горизонтали и от 0 до dy по вертикали. Тогда проверочные условия примут вид:

CASE rt$: x = x0 + 5: IF x + dx > xmax THEN x = x0

CASE dn$: y = y0 + 5: IF y + dy > ymax THEN y = y0

   Естественно значения dx и dy задавать после операторов DATA, а переменные xmax и ymax инициализировать после оператора SCREEN. В результате доработок получим следующую программу:

‘ ---------- Программа “Рожица” ----------


‘Задаем картинку
DATA 0, 0, 7, 7, 7, 7, 7, 0, 0

DATA 0, 8, 0, 0, 0, 0, 0, 8, 0

DATA 8, 0, 0, 8, 0, 8, 0, 0, 8

DATA 8, 0, 3, 3, 0, 3, 3, 0, 8

DATA 8, 0, 0, 0, 0, 0, 0, 0, 8

DATA 8, 0, 0, 12, 12, 12, 0, 0, 8

DATA 0, 8, 0, 0, 0, 0, 0, 8, 0

DATA 0, 0, 8, 8, 8, 8, 8, 0, 0

dx = 8: dy = 7

‘Выводим ее на экран
SCREEN 7: xmax = 319: ymax = 199

DIM a%(40)

FOR y = 0 TO dy

   FOR x = 0 TO dx

      READ c: PSET (x, y), c

   NEXT x

NEXT y

‘Запоминаем картинку в массиве и очищаем экран
GET (0, 0)-(dx, dy), a%

CLS

‘Задаем исходное положение картинки и выводим ее на экран
x0 = 165: y0 = 45: x = x0: y = y0

PUT (x0, y0), a%

‘Ждем нажатия клавиши
10 w$ = INKEY$: IF w$ = "" THEN GOTO 10

‘Вычисляем новое положение картинки
SELECT CASE w$

   CASE   lf$: x = x0 - 5: IF x < 0 THEN x = x0

   CASE  up$: y = y0 - 5: IF y < 0 THEN y = y0

   CASE  rt$: x = x0 + 5: IF x +  dx > xmax THEN x = x0

   CASE  dn$: y = y0 + 5: IF y + dy > ymax THEN y = y0

   CASE CHR$(27): STOP

   CASE ELSE: GOTO 10

END SELECT

‘Стираем картинку на старом и выводим на новом месте
PUT (x0, y0), a%

PUT (x, y), a%

‘Обновляем переменные
x0 = x: y0 = y

GOTO 10

   Задание. Запишите программу под тем же именем. Нарисуйте фон, по которому будет двигаться рожица. Убедитесь, что при движении рожицы фон не портится. Задайте вместо рожицы другую картинку (человечка, кошку, крестик и т.д.). Сохраните результат под другим именем.

 

14. Графический редактор “Раскрашка”

   Применим полученные знания для выполнения довольно большого проекта - написания графического редактора “Раскрашка”. В этом редакторе можно закрашивать заложенные в программу контурные рисунки. Опишем схему работы программы.

   На экране находится контурный рисунок и палитра цветов. Для выбора цвета закраски передвигаем курсор - картинку-указатель на прямоугольник с нужным цветом и нажимаем Enter. Затем перемещаем курсор на основное поле так, чтобы он указывал на какую-нибудь область рисунка. После нажатия на Enter область закрашивается выбранным ранее цветом. Для правильной работы оператора закрашивания нужно, чтобы все границы были нарисованы одним и тем же цветом. Понятно, что этот цвет следует исключить из палитры, так как область, закрашенную цветом границы, нельзя будет перекрасить другим цветом.

   За основу “Раскрашки” возьмем программу “Рожица”. Загрузим программу и запишем ее в файл RASCRAS.BAS куда и будем вносить все изменения.

   Начнем с курсора. Зададим шаблон курсора-стрелки, отмечая точки курсора единичками.

‘Шаблон курсора-стрелки

 DATA 1, 0, 0, 0, 0, 0, 0, 0

 DATA 1, 1, 0, 0, 0, 0, 0, 0

 DATA 1, 1, 1, 0, 0, 0, 0, 0

DATA 1, 1, 1, 1, 0, 0, 0, 0

DATA 1, 1, 1, 1, 1, 0, 0, 0

DATA 1, 1, 1, 1, 1, 1, 0, 0

DATA 1, 1, 1, 1, 1, 1, 1, 0

DATA 1, 1, 1, 1, 1, 1, 1, 1

DATA 1, 1, 1, 1, 1, 0, 0, 0

DATA 1, 1, 1, 1, 1, 0, 0, 0

DATA 1, 1, 1, 1, 1, 1, 0, 0

DATA 1, 0, 0, 0, 1, 1, 0, 0

DATA 0, 0, 0, 0, 1, 1, 0, 0

DATA 0, 0, 0, 0, 1, 1, 1, 0

DATA 0, 0, 0, 0, 0, 1, 1, 0

DATA 0, 0, 0, 0, 0, 1, 1, 0

Как и раньше, выведем курсор на экран для дальнейшего запоминания в памяти, после чего экран очистим. Но цвет 1 - это синий цвет, и стрелка будет плохо видна на экране. Поэтому при выводе заменим цвет курсора на более яркий.

dx = 7: dy = 15

SCREEN 7: xmax = 319: ymax = 199

DIM a%(40)

FOR y = 0 TO dy

   FOR x = 0 TO dx

      READ c: IF c = 1 THEN c = 10  ‘Изменение цвета курсора
      PSET (x, y), c

   NEXT x

NEXT y

‘Запоминаем картинку в массиве и очищаем экран
GET (0, 0)-(dx, dy), a%

CLS

   Теперь рисуем рамку основного поля, палитру и индикатор.

LINE (0, 0)-(xmax, ymax), 15, B

LINE (80, 0)-(80, ymax), 15

LINE (10, 10)-(70, 160), 15, B<

‘Заполняем палитру красками
c = 14

FOR y = 20 TO 150 STEP 10

   LINE (10, y)-(80, y), 15

   PAINT (40, y-5), c, 15

   c = c - 1

NEXT y

‘Индикатор
CIRCLE (40,180), 10, 15

   Далее должна следовать часть программы, создающая контурный рисунок на основном поле. Нарисуем, например, “мишень”:

FOR r = 10 TO 100 STEP 15

   CIRCLE (200, 100), r, 15

NEXT r

   Чтобы контурный рисунок не мог нечаянно выйти за пределы основного поля, можно ограничить область вывода с помощью оператора VIEW:

   VIEW (x1, y1)-(x2, y2) - устанавливает прямоугольное окно вывода с новой системой координат, начало которой располагается в левом верхнем углу окна. Относительные (локальные) координаты позволяют не задумываться о том, в каком месте экрана будет в дальнейшем расположен рисунок. Если по каким-либо причинам требуется сохранить в окне систему координат экрана, то следует использовать вариант 

   VIEW SCREEN (x1, y1)-(x2, y2).

   Чтобы восстановить возможность вывода на весь экран, следует применить оператор VIEW без параметров.

   Следующую часть программы - управление курсором - позаимствуем из программы “Рожица”, только в комментариях заменим “картинку” на “курсор”. Изменим и шаг перемещения курсора, чтобы можно было закрашивать маленькие области. (Оптимальный вариант - ввести переменную, значение которой можно было бы изменять, нажимая какие-то клавиши на клавиатуре, например, клавиши“<” и “>”). 

‘Задаем константы
lf$ = CHR$(0) + CHR$(75)

rt$ = CHR$(0) + CHR$(77)

up$ = CHR$(0) + CHR$(72)

dn$ = CHR$(0) + CHR$(80)

‘Задаем исходное положение курсора и выводим его на экран
x0 = 165: y0 = 45: x = x0: y = y0

PUT (x0, y0), a%

‘Ждем нажатия клавиши
10 w$ = INKEY$: IF w$ = "" THEN GOTO 10

‘Вычисляем новое положение курсора
SELECT CASE w$

   CASE   lf$: x = x0 - 2: IF x < 0 THEN x = x0

   CASE  up$: y = y0 - 2: IF y < 0 THEN y = y0

   CASE  rt$: x = x0 + 2: IF x +  dx > xmax THEN x = x0

   CASE  dn$: y = y0 + 2: IF y + dy > ymax THEN y = y0

   CASE CHR$(27): STOP

   CASE ELSE: GOTO 10

END SELECT

‘Стираем курсор на старом и выводим на новом месте
PUT (x0, y0), a%

PUT (x, y), a%

‘Обновляем переменные
x0 = x: y0 = y

GOTO 10

   Добавим теперь в оператор выбора реакцию программы на нажатие клавиши Enter. Эту клавишу мы нажимаем в двух случаях: когда хотим выбрать цвет закрашивания, и когда указываем область для закраски. Различить эти случаи легко: в первом случае острие стрелки (точка x0, y0) находится в прямоугольнике (10, 10)-(70, 160) - это границы палитры; во втором - в прямоугольнике (80, 0)-(xmax, ymax) - это границы основного поля.

   CASE CHR$(13)

      IF x0 > 10 AND x0 < 70 AND y0 > 10 AND y0 < 160 THEN

        ‘Определяем цвет краски на палитре
         c = 15 - y0 \ 10

        ‘Прячем курсор
         PUT (x0, y0), a%

        ‘закрашиваем индикатор
         PAINT (40, 180), c, 15

        ‘и восстанавливаем курсор на том же месте
         PUT (x0, y0), a%

      END IF

   Получить формулу, которая связывает номер цвета с расположением краски на палитре, поможет следующая таблица:

Интервал

Номер
интервала

Цвет

10 < y < 20

1

14

20 < y < 30

2

13

30 < y < 40

3

12

...

...

...

150<y<160

15

0

   Легко видеть, что номер интервала и цвет в сумме равны 15 в любой строке таблицы, а номер интервала равен количеству полных десятков в числе y. Вычислить номер интервала можно с помощью операции целочисленного деления: n = y \ 10. После этого цвет краски получается по формуле c = 15 - n.

   Заметим, что простота формулы есть результат простой закономерности в расположении цветов палитры. В других случаях может оказаться проще “считывать” цвет непосредственно с экрана, используя функцию POINT.

   POINT(x, y) - возвращает цвет пикселя экрана с координатами (x, y).

   Зачем потребовалось стирать курсор перед закрашиванием индикатора? Дело в том, что при закраске области может оказаться, что курсор задевает ее границу. В этом месте граница исчезнет, и через образовавшуюся “дыру” краска вытечет за пределы области. Поэтому перед любым закрашиванием мы будем убирать курсор с экрана, а после - возвращать на экран. (“Прятать” курсор придется и при определении функцией POINT цвета точки (x0, y0), которая находится под курсором.)

   Займемся теперь основным полем.

      IF x0 > 80 AND x0 < xmax AND y0 > 0 AND y0 < ymax THEN

        ‘Прячем курсор
         PUT (x0, y0), a%

        ‘Закрашиваем область
         PAINT (x0, y0), c, 15

        ‘Восстанавливаем курсор
         PUT (x0, y0), a%

      END IF

   Задание. Придумать и вставить в программу свой рисунок для раскрашивания (ниже приведен пример выполнения задания). 

15. Графические режимы монитора VGA 

   При наличии монитора VGA дополнительно можно пользоваться графическими режимами SCREEN 10, 11, 12 или 13. 

   SCREEN 10 - разрешение 640х350 точек, цветовая палитра - черно-белая с возможностью мигания ( 0 - черный цвет, 1 - белый цвет, 2 - мигающий ). 

   SCREEN 11 - разрешение 640х480 точек, цветовая палитра - черно-белая ( 0 - черный цвет, 1 - белый цвет). 

   SCREEN 12 - разрешение 640х350 точек, цветовая палитра - 16 цветов. В этом и следующем (SCREEN 13) режиме оператор COLOR имеет несколько другой синтаксис:

   COLOR цвет_рисования - устанавливает цвет “по умолчанию” для графических операторов PSET, LINE, CIRCLE и др.

   SCREEN 13 - разрешение 320х200 точек, цветовая палитра может содержать до 256 цветов (в Quick Basic - 248). По количеству точек этот режим совпадает с режимом SCREEN 7, но каждая точка может быть окрашена в один из 248 цветов. 

   Познакомимся поближе с последним режимом. Если в программе “Раскрашка” заменить SCREEN 7 на SCREEN 13, то при выполнении оператора 

GET (0, 0)-(dx, dy), a%

появится сообщение “Illegal function call”, что означает, что курсор не помещается в массив a%. А почему? Дело в том, что для задания любого из 16 цветов необходимо 4 бита информации (16 = 24), а для задания 256 цветов - 8 бит (256 = 28), т.е. при работе в режиме SCREEN 13 массив a% должен быть вдвое больше, чем в SCREEN 7. После этого программа заработает, но в ней все останется по-прежнему! Чтобы изменить палитру цветов нужно взять цвета с номерами, отличными от 0 - 15. Сразу же за этими номерами идет палитра оттенков серого цвета - от цвета номер 16 (черного) до цвета номер 31 (белого). Изменим начальное значение с = 14 на с = 31 (Quick Basic различает цвета с номерами 31 и 15, хотя “по виду” они одинаковы!) и перепишем фрагмент программы для определения цвета закрашивания:

        ‘Определяем цвет краски на палитре
        ‘Прячем курсор,
         PUT (x0, y0), a%

        ‘определяем цвет,
         c = POINT(x0, y0)

        ‘закрашиваем индикатор
         PAINT (40, 180), c, 15

        ‘и восстанавливаем курсор на том же месте
         PUT (x0, y0), a%

   На самом деле, здесь мы допустили небольшую ошибку - не учли, что курсор может показать на белую линию, разделяющую цвета в палитре. И хотя при запуске программа будет работать безошибочно, то только потому, что начальное положение курсора и шаг перемещения не позволят курсору “встать” на разделительную линию. При дальнейшей модификации программы скрытая ошибка может обнаружить себя. Поэтому, исправим ее теперь:

        ‘Определяем цвет краски на палитре
        ‘Прячем курсор,
         PUT (x0, y0), a%

        ‘определяем цвет,
         c0 = POINT(x0, y0)

         IF c0 <> 15 THEN

            c = c0

           ‘закрашиваем индикатор
            PAINT (40, 180), c, 15

         END IF

        ‘и восстанавливаем курсор на том же месте
         PUT (x0, y0), a%

   Цвета, идущие за палитрой серого цвета, можно разбить на 9 палитр по 24 цвета. Сходство цветов в палитрах хорошо видно, если их вывести на экран параллельными полосками. 

   Задание. Напишите программу для вывода на экран всех палитр в режиме SCREEN 13. Подберите свою палитру цветов для программы “Раскрашка” и вставьте ее в программу с помощью оператора DATA. Измените программу так, чтобы она при рисовании палитры “читала” цвет из списка данных.


